

**R. Brigola, TH Nürnberg Georg Simon Ohm, 2013**

**Mathematica - Notebooks als Bonusmaterial zum Lehrbuch**

**[1] Rolf Brigola Fourier-Analysis und Distributionen,  
Eine Einführung mit Anwendungen,  
edition swk, Hamburg 2013**

**Teil 2 Erste Anwendungsbeispiele von Fourierreihen**

Referenzen zu Kapiteln, Abschnitten, Seitenzahlen beziehen sich im Folgenden meist auf das genannte Lehrbuch. Einige wenige weitere Referenzen findet man am Ende des Notebooks.

Ziel dieses zweiten Notebooks zum Thema ist, Nutzern einige erste Beispiele von Anwendungen von Fourierreihen mit Hilfe von *Mathematica* vorzustellen. Das Notebook ist mit *Mathematica* 9 unter Windows 7 erstellt. Dieses und weitere Demo-Notebooks findet man unter folgender URL des Autors: [www.stiftung-swk.de/mathematica](http://www.stiftung-swk.de/mathematica)

### **2.1 Asymptotisch stabile lineare Differentialgleichungen mit konstanten Koeffizienten und stetiger, stückweise stetig differenzierbarer periodischer Inhomogenität**

Asymptotisch stabile lineare gewöhnliche Differentialgleichungen  $P(D)u=f$  von der Ordnung  $n$  mit konstanten Koeffizienten besitzen ein charakteristisches Polynom  $P(z) = \sum_{k=0}^n a_k z^k$  ( $D$  steht oben für den Differentialoperator  $d/dt$ ), dessen Nullstellen sämtlich negative Realteile haben. Notwendig ist, dass  $P$  ein sog. Hurwitz-Polynom ist, d.h. dass alle Koeffizienten gleiches Vorzeichen besitzen. In [1], Kapitel 5, Abschnitt 2 wird dargelegt, dass solche Gleichungen mit einer stetigen, stückweise stetig differenzierbaren  $T$ -periodischen rechten Seite  $f$  eine eindeutig bestimmte  $T$ -periodische Lösung  $u$  haben, die man durch die  $T$ -periodische Faltung von  $f$  mit der entsprechend  $T$ -periodischen Übertragungsfunktion  $h$  erhält. Die Fourierreihe dieser  $T$ -periodischen Übertragungsfunktion hat die Fourierkoeffizienten  $h_k = 1/P(i\omega_0 k)$ ,  $i^2 = -1$ ,  $\omega_0 = 2\pi/T$ . Die Faltung ist  $n$ -mal stetig differenzierbar. Sie beschreibt das Langzeitverhalten der Lösung nach dem (schnellen) Abklingen des Einschwingvorgangs und hat üblicherweise ein anderes Amplituden- und Phasen-Spektrum als die Anregung  $f$ . In Abschnitt 8.1 von [1] wird gezeigt, dass dieser Faltungszusammenhang von  $u$  und  $f$  unter Aufgabe von Differenzierbarkeitseigenschaften von  $f$  ganz allgemein auch für periodische Distributionen  $f$  gilt. Die Lösung  $u$  ist dann ebenfalls als periodische Distribution aufzufassen. (Man vgl. hierzu [1], Abschnitt 10.5, S. 288-289.)

### **Wir betrachten als Beispiel eine erzwungene Schwingung eines Federpendels**

Ein Federpendel der Masse  $m$  werde mit der Kraft  $F$  angeregt. Der Dämpfungskoeffizient sei  $k>0$ , die Federkonstante sei  $d>0$ . Die beschreibende Differentialgleichung für die Auslenkung  $y[t]$  lautet:

In[1]:=

```
ClearAll["Global`*"]; Remove["Global`*"]; ?Global`*
```

Remove::rmnsm : There are no symbols matching "Global`\*". >>

Information::nomatch : No symbol matching Global`\* found. >>

In[2]:=  $\text{dgl} = m y''[t] + k y'[t] + d y[t] == F[t]$ Out[2]:=  $d y[t] + k y'[t] + m y''[t] == F[t]$ 

### a) Anregende Kraft eine einfache Cos-Schwingung

Wir setzen zur Veranschaulichung konkret  $F(t) = F_0 \cos[\omega t]$ ,  $F_0 = 0.2 \text{ N}$ ,  $m = 1 \text{ kg}$ ,  $k = 1/5 \text{ kg/s}$ ,  $d = 1/2 \text{ N/m}$ ,  $\omega = 2\pi \text{ s}^{-1}$ . Die Gleichung ist asymptotisch stabil, wie man sofort an den Nullstellen des charakteristischen Polynoms erkennt: Es hat konjugiert komplexe Nullstellen mit negativen Realteilen.

```
In[3]:= m := 1; k := 1/5; d := 1/2; F0 := 0.2; ω := 2π; F[t_] := F0 Cos[ω t];
P[z_] := m z^2 + k z + d;
Nullstellen = Solve[P[z] == 0, z]
```

```
Out[5]:= {{z -> -1/10 - 7i/10}, {z -> -1/10 + 7i/10}}
```


Nachfolgend die **periodische Lösung**, die man erhält, wenn man die Lösung der homogenen Differentialgleichung Null setzt, hier indem die auftretenden Parameter  $C[1]$ ,  $C[2]$  Null gesetzt werden. **Diese Lösung hat die gleiche Frequenz wie die anregende Kraft  $F$  mit der Schwingungsdauer  $T=1\text{s}$ , aber eine andere Phase und eine andere, kleinere Amplitude.** Sie beschreibt die resultierende Schwingung nach Abklingen des Einschwingvorgangs. Wir lösen die Gleichung zunächst mit den vorhandenen *Mathematica*-Algorithmen.

In[6]:=  $\text{solution} := \text{DSolve}[\text{dgl}, y, t]$ 

Die periodische Lösungsfunktion nennen wir  $f$ , die nachfolgend mit dem *Mathematica*-Ergebnis definiert und ausschnittsweise grafisch dargestellt wird.

In[7]:=  $f = \text{solution}[[1, 1, 2]] /. \{C[1] \rightarrow 0, C[2] \rightarrow 0\};$ 

```
In[8]:= Plot[f[t], {t, 0, 6}, PlotStyle -> Directive[
Blue, Thickness[0.005]], PlotRange -> All]
```



Die resultierende Amplitude und Phasenverschiebung erhält man einfach aus dem Frequenzgang. Die Fourierkoeffizienten der Anregung  $F$  sind im vorliegenden Fall mit  $1/P(\pm i\omega)$  zu multiplizieren.

Wir erzeugen zunächst die Fourierkoeffizienten von  $F$  und damit noch einmal die Lösung, die man durch **periodische Faltung von  $F$  mit der entsprechend periodischen Übertragungsfunktion** gewinnt (vgl. [1], Abschnitt 5.2), also durch Multiplikation der zugehörigen Fourierkoeffizienten. Wie bei allen zeitinvarianten linearen Systemen werden "keine neuen Frequenzen" erzeugt. Wir geben die Lösung noch einmal mit unseren vorhandenen Kenntnissen über Fourierreihen an und nennen das Ergebnis diesmal  $g$ :

In[9]=

```
FourierCoefficient[F[t], t, n, FourierParameters -> {1, 2 Pi}]
```


Out[9]=  $\begin{cases} 0.1 & n == -1 \mid \mid n == 1 \\ 0. & \text{True} \end{cases}$

Nun das in diesem Fall trigonometrische Polynom, das die Gleichung löst und noch einmal der zugehörige Plot:

In[10]=  $g[t_] = \text{FullSimplify}[1 / (10 P[-i \omega]) \text{Exp}[-i \omega t] + 1 / (10 P[i \omega]) \text{Exp}[i \omega t]]$

Out[10]=  $\left( (10 - 80 \pi^2) \text{Cos}[2 \pi t] + 8 \pi \text{Sin}[2 \pi t] \right) / \left( 25 + 64 \pi^2 (-6 + 25 \pi^2) \right)$

In[11]=  $\text{Plot}[g[t], \{t, 0, 6\}, \text{PlotStyle} \rightarrow \text{Directive}[$   
 $\text{Blue}, \text{Thickness}[0.005]], \text{PlotRange} \rightarrow \text{All}]$



Nun resultierende Amplitude und Phasenverschiebung, berechnet aus dem Frequenzgang mit  $1/P(i \omega)$ ;  $\omega$  war die Kreisfrequenz der Erregung.

In[12]=  $\text{amp} := 2 \text{Abs}[\text{FourierCoefficient}[F[t], t, 1, \text{FourierParameters} \rightarrow \{1, 2 \text{Pi}\}] / P[i \omega]]$   
 $\text{N}[\text{amp}]$

Out[13]= 0.00512838

In[14]=  $\text{phase} := \text{Arg}[1 / P[i \omega]]$   
 $\text{N}[\text{phase}]$

Out[15]= -3.10936

Die positive **Resonanzkreisfrequenz** und die Lösungsamplitude bei einer solchen Frequenz ergibt sich aus der Maximalstelle von  $|1/P(i\omega)|$ ,  $\omega > 0$ . Sie lautet

$\omega_r = \sqrt{d/m - k^2/(2m^2)}$ . Man kann auch *Mathematica* nach dieser Extremstelle

mit dem Befehl **FindMaximum** suchen lassen. Vergleichen Sie den Wert am weiter unten dargestellten Amplitudengang. Da unsere Anregungsfrequenz weit oberhalb der Resonanzfrequenz lag, beobachten wir eine **sehr starke Amplitudendämpfung bei der resultierenden Schwingung**.

Die der Form nach selbe Differentialgleichung beschreibt in der Regelungstechnik ein

sogenanntes **PT<sub>2</sub>-Glied**, oder anders gesagt ein **Tiefpassfilter 2. Ordnung**.

**Übung:** Testen Sie die Aussage durch Veränderung der Anregungsfrequenz und passende FourierParameters, Periodendauer etc.

```
In[16]:= resonanzkreisfrequenz = Sqrt[d / m - k^2 / (2 m^2)]
```

```
Out[16]=  $\frac{2 \sqrt{3}}{5}$ 
```

```
In[17]:= N[resonanzkreisfrequenz]
```

```
Out[17]= 0.69282
```

Die Amplitude der Lösung bei dieser Anregungskreisfrequenz wäre

```
In[18]:= N[2 Abs[FourierCoefficient[F[t], t, 1,
FourierParameters -> {1, 2 Pi}] / P[i resonanzkreisfrequenz]]]
```


```
Out[18]= 1.42857
```

```
In[19]:= FindMaximum[Abs[1 / P[I w]], w, WorkingPrecision -> 30]
```


```
Out[19]= {7.14285714285714285714285714277, {w -> 0.692820323027550416899176753060}}
```

Nachfolgend noch der Amplitudengang und der Phasengang des als lineares Übertragungssystem aufgefassten Pendels. Eine Darstellung jeweils nur für positive Kreisfrequenzen ist aufgrund der bekannten Symmetrien ausreichend (vgl. [1]).

```
In[20]:= Plot[Abs[1 / P[i w]], {w, 0, 6.3}, PlotStyle -> Directive[
Red, Thickness[0.005]], PlotRange -> All]
```



```
In[21]:= Plot[Arg[1 / P[i ω]], {ω, 0, 6.3}, PlotStyle → Directive[
  Red, Thickness[0.005]], PlotRange → All]
```



Hier als letzter Test noch Einsetzen in die Differentialgleichung:  
Bis auf numerische Rundungsfehler, die den nachfolgenden Sinus-Term erzeugen,  
erhalten wir wieder die rechte Seite  $F[t]$ :

```
In[22]:= test[t_] = amp Cos[ω t + phase];
FullSimplify[m test''[t] + k test'[t] + d test[t]]
```

```
Out[23]= 0.2 Cos[2 π t] + 2.60209 × 10-18 Sin[2 π t]
```

### b) Eine Fourierreihe als periodische Anregung

Wir betrachten nun das gleiche Problem mit  $F[t] = F_0 \sum_{k=1}^{\infty} \text{Cos}[3 k t] / k^2 = F_0 \sum_{k=-\infty, k \neq 0}^{+\infty} \text{Exp}[i 3 k t] / (2 k^2)$ .

Die von der Reihe dargestellte Funktion ist die  $2\pi/3$ -periodische Fortsetzung der auf  $[0, 2\pi/3[$  definierten Funktion  $f[t] = F_0 ((3 t - \pi)^2 / 4 - \pi^2 / 12)$  (vgl. [1], S. 29). Sie ist stetig und stückweise stetig differenzierbar.

Mit Hilfe der periodischen Übertragungsfunktion mit den Fourierkoeffizienten  $1/P(i 3 k)$  (die Grundkreisfrequenz ist jetzt  $3 \text{ s}^{-1}$ , die zugehörige Schwingungsdauer  $2\pi/3 \text{ s}$ ) kann man die periodische Lösung  $u$  in Form einer Fourierreihe sofort angeben: Sie lautet

$$u[t] = \sum_{k=-\infty, k \neq 0}^{+\infty} F_0 e^{i 3 k t} / (2 k^2 P[i 3 k]).$$

Zur Veranschaulichung und Ersparnis überlanger Rechenzeit begnügen wir uns mit einer Partialsumme der Reihe für die Kraft als Näherung und einem entsprechenden trigonometrischen Polynom als Näherung für die Lösung. Wir plotten über drei Perioden der Anregung (blau) und der Lösung (rot). Die Amplitudendämpfungen und Phasenverzögerungen aller Teilschwingungen wirken in der resultierenden Pendelbewegung zusammen und wie üblich ist die Lösung durch die Glättung, die mathematisch durch die periodische Faltung bewirkt wird, "weniger eckig" als die Anregung und hat bedeutend weniger Leistung als diese (vgl. [1], S. 50-51). Physikalisch gesprochen liegt die Ursache dieser Glättung in der Massenträgheit: Die Masse kann aufgrund ihrer Trägheit schon nach wenigen Oberschwingungen der Anregungsfrequenz nicht mehr folgen. Daher sind für physikalische Betrachtungen oft schon wenige Glieder der Fourierreihen-Lösung bei der Analyse gedämpfter Schwingungen wie im Beispiel ausreichend, um sehr gute Näherungen der "exakten" Lösung zu erhalten. Man vergleiche hierzu einfach einmal die Näherungslösungen mit nur 5 wie nachfolgend und bis zu 100 oder 500 Summanden einer Partialsumme dieser Lösung.


```
In[24]:= F[t_] := F0 Sum[Exp[-i n 3 t] / (2 n^2) + Exp[i n 3 t] / (2 n^2), {n, 1, 5}]
```

```
In[25]:= u[t_] :=
  F0 Sum[Exp[-i n 3 t] / (2 n^2 P[-i 3 n]) + Exp[i n 3 t] / (2 n^2 P[i 3 n]), {n, 1, 5}]
```

```
In[26]:= g1 := Plot[u[t], {t, 0, 2 π}, PlotStyle → Directive[
  Red, Thickness[0.005]], PlotRange → All]
```

```
In[27]:= g2 := Plot[F[t],
  {t, 0, 2 π}, PlotStyle → Directive[Blue, Thickness[0.005]], PlotRange → All]
```

```
In[28]:= Show[g1, g2]
```



Nachfolgend zum Vergleich die quadratischen Mittelwerte (Quadrate der  $L^2$  – Normen, Leistungen) der Anregung und der resultierenden Lösungsfunktion, der die Leistungsverluste verdeutlicht :

```
In[29]:= NIntegrate[Abs[F[t]]^2 / (2 π / 3), {t, 0, 2 π / 3}]
```

Out[29]= 0.021607

```
In[30]:= NIntegrate[Abs[u[t]]^2 / (2 π / 3), {t, 0, 2 π / 3}]
```

Out[30]= 0.000276477

Alle hier beispielhaft für das Pendel durchgeführten Überlegungen und Rechnungen lassen sich vollkommen analog auch für **stabile zeitinvariante lineare Übertragungssysteme** anwenden, die durch Differentialgleichungen wie oben (auch höherer Ordnungen) beschrieben werden. In der Elektrotechnik sind sie die Grundlage, die ein Fach wie die **“Wechselstromlehre”** mit den Begriffen des **Frequenz-, des Amplituden- und Phasengangs** und daraus abgeleiteter Größen wie die **“Gruppenlaufzeit”** überhaupt erst sinnvoll macht, da reale Signale kaum jemals eine reine Sinus-Form haben, sich aber, wenn sie zeitbegrenzt in einem Zeitintervall  $[0, T]$  sind, mit der Theorie von Fourierreihen verstehen und behandeln lassen, wenn man sie sich T-periodisch fortgesetzt denkt. Fourierreihen sind daher ein wertvolles theoretisches Hilfsmittel zur Beschreibung einer großen Anzahl von Vorgängen in der Technik.

**Zur selbständigen Übung empfehle ich Ihnen, eigene Beispiele und die Aufgaben der Kapitel 3 - 5 in [1] zu rechnen und mit *Mathematica* auf Richtigkeit Ihrer Ergebnisse zu prüfen.**

## 2.2 Lösung der eindimensionalen Wärmeleitungsgleichung für einen Stab mit homogenen und mit Neumannschen Randbedingungen

Wir beginnen mit der homogenen Wärmeleitungsgleichung mit homogenen Randbedingungen. Das Anfangsrandwertproblem für die homogene Wärmeleitungsgleichung für einen (thermisch dünnen) Stab der Länge  $L$ , dessen Enden eisgekühlt sind, beschrieben als Intervall  $[0, L]$  und eine Anfangstemperatur  $f[x]$  in  $x \in [0, L]$ , lautet mit dem Koeffizienten  $a$  für die Temperaturleitfähigkeit (siehe etwa C. Gerthsen, H. Kneser Physik u.a.).

$$\partial_t u[x, t] = a \partial_{x,x} u[x, t], \quad u[x, 0] = f[x], \quad u[0, t] = u[L, t] = 0 \quad \text{für alle } t \geq 0.$$

Mit einem Separationsansatz  $u[x, t] = v[x] w[t]$  erhält man - ganz analog zur Lösung der Differentialgleichung für die schwingende Saite in [1], S. 2-4 und Aufgabe A8 zu Kap. 5) - die Fourierreihenlösung

$$u[x, t] = \sum_{n=1}^{\infty} b_n e^{-\lambda_n^2 t} \text{Sin}[n\pi x / L]$$

mit den Fourier-Sinusoeffizienten der  $2L$ -periodischen ungeraden Fortsetzung von  $f$  und  $\lambda_n = n\pi\sqrt{a}/L$ . Dabei setzen wir voraus, dass  $f$  stetig und stückweise stetig differenzierbar ist und damit eine gleichmäßig konvergente Fourierreihe besitzt. Zur Demonstration wählen wir  $L=1$  m und  $f$  als Parabelbogen

$f[x] = 5x(L-x)$  in  $C^0$  und die Temperaturleitfähigkeit  $a = 117 \cdot 10^{-6} \text{ m}^2/\text{s}$  von Kupfer (bei etwa  $20^\circ \text{C}$ ).

Nachfolgend berechnen wir die Fourierkoeffizienten von  $f$  und plotten eine Näherung der Funktion  $u$  als 3D-Grafik, die den zeitlichen Verlauf des Temperatursausgleichs im Stab für  $t \geq 0$  veranschaulicht.

Vor dem Start löschen wir alle bisher vergebenen Symbole und Variablen.

```
In[31]:= ClearAll["Global`*"]; Remove["Global`*"]; ?Global`*
```

Information::nomatch : No symbol matching Global`\* found. >>


```
In[32]:= L := 1; f[x_] := 40 x (L - x); a := 117 10^(-6);
b[n_] = 2 / L Integrate[f[x] Sin[n Pi x / L], {x, 0, L}]
```

```
Out[33]:= - \frac{1}{n^3 \pi^3} 80 (-2 + 2 \text{Cos}[n \pi] + n \pi \text{Sin}[n \pi])
```

Hier eine Fourierreihen-Näherung für die Lösung als trigonometrisches Polynom vom Grad  $m$  und eine Darstellung mit  $m=4$ . Sie fällt mit wachsender Zeit auf Null ab:

```
In[34]:= u[m_, x_, t_] := Sum[b[n] Exp[-(n Pi Sqrt[a] / L)^2 t] Sin[n Pi x / L], {n, 1, m}]
```

```
In[35]= Plot3D[u[4, x, t], {x, 0, L}, {t, 0, 3600},
  PlotRange -> All,
  AxesLabel -> {" Ort x in m ", "Zeit t in s ", "Temperatur u in C0 "},
  AxesStyle -> Directive[Black, Bold, 12], ViewPoint -> {-2, -2, 3}]
```



```
In[36]= ClearAll["Global`*"]; Remove["Global`*"]; ?Global`*
```

Information::nomatch : No symbol matching Global`\* found. >>

Nun zum gleichen Problem mit **Neumannschen Randbedingungen**, d.h. physikalisch bei isolierten Stabenden, so dass keine Wärme über den Rand fließt, mathematisch  $\partial_x T[0, t] = \partial_x T[L, t] = 0$  für den Temperaturverlauf  $T[x, t]$ .

Wir lösen nun das Problem schrittweise analog zum Vorgehen, das hinter der Lösung (zur Eigenarbeit) im vorherigen Beispiel steht. Der Separationsansatz  $u[x, t] = v[x] w[t]$  führt auf zwei gewöhnliche Differentialgleichungen  $v'' + cv = 0$  und  $w' + \alpha cw = 0$  und  $v'[0] = 0, v'[L] = 0$  (die Temperaturleitfähigkeit ist nun mit  $\alpha$  bezeichnet, die Temperatur mit  $T$ ):

Mit einem Separationsansatz, Einsetzen in die Gleichung und die Randbedingungen erhalten wir:

$$\partial_t T[x, t] - \alpha \partial_{x,x} T[x, t] = 0 \text{ und } \partial_x T[0, t] = \partial_x T[L, t] = 0 :$$

```
In[37]= sol1 = DSolve[{v''[x] + c v[x] == 0}, v[x], x]
```

```
Out[37]= {{v[x] -> C[1] Cos[sqrt[c] x] + C[2] Sin[sqrt[c] x]}}
```

```
In[38]= v[x_] = sol1[[1, 1, 2]]
```

```
Out[38]= C[1] Cos[sqrt[c] x] + C[2] Sin[sqrt[c] x]
```

Wir setzen die Randbedingung in den Lösungsanteil  $v$  ein und erhalten aus  $v'[0] = 0$ , dass  $C[2] = 0$  sein muss, falls  $c > 0$  ist. (Für  $c < 0$  gibt es nur die triviale Lösung.)

```
In[39]= v'[0]
```

```
Out[39]= sqrt[c] C[2]
```

Es bleibt  $v'[L] = -C[1] \sqrt{c} \sin[\sqrt{c} L] = 0$ . Die resultierenden Möglichkeiten sind Werte für die Konstante  $c$  der Form  $c_n = n^2 \pi^2 / L^2$  für  $n$ , die alle Lösungen ergeben.



In[40]:= `Solve[Sqrt[c] L == n Pi, c]`

$$\text{Out[40]} = \left\{ \left\{ c \rightarrow \frac{n^2 \pi^2}{L^2} \right\} \right\}$$

**Erstes Teilergebnis:** Eine Folge von möglichen Lösungen  $v_n[x]$ ,  $n=1,2, \dots$

$v_n[x] = a_n \text{Cos}\left[\sqrt{c_n} x\right]$ , die alle unsere Randbedingung erfüllen. Die  $a_n$  sind noch frei wählbare Konstanten. Nun zur Lösung für den zweiten Teil des Separationsansatzes :

In[41]:= `sol2 = DSolve[w'[t] + \alpha n^2 Pi^2 / L^2 w[t] == 0, w[t], t]`

$$\text{Out[41]} = \left\{ \left\{ w[t] \rightarrow e^{-\frac{n^2 \pi^2 t \alpha}{L^2}} C[1] \right\} \right\}$$

Hiermit erhalten wir nun insgesamt eine Fourierreihen-Lösung durch Superposition und notieren einen Ausschnitt der Lösungsreihe vom Grad  $m_2$  (eine konstante Funktion  $T[x,t]=c_0$  ist ebenfalls eine Lösung)

$$T[m1_, m2_, x_, t_] = c_0 + \text{Sum}\left[ a[n] \text{Exp}\left[-\frac{\alpha n^2 \pi^2 t}{L^2}\right] \text{Cos}[n \pi x / L], \{n, m1, m2\}\right]$$

Die  $a[n]$  müssen also nach der gegebenen Anfangsbedingung gerade die Fourier - Cosinusoeffizienten der Anfangsbedingung  $f$  sein,  $c_0$  ist der Mittelwert von  $f$ . Mit den gleichen Daten wie im ersten Beispiel berechnen und zeichnen wir den näherungsweisen Temperaturverlauf mit der Partialsumme vom Grad 8. Mit variierenden Parametern  $m_1, m_2$  können Sie sich die Entwicklung von Abschnitten der Reihe ansehen.


In[42]:= `L := 1; f[x_] := 40 x (L - x); \alpha := 117 10^(-6);  
a[n_] = 2 / L Integrate[f[s] Cos[n Pi s / L], {s, 0, L}]  
c0 = 1 / L Integrate[f[s], {s, 0, L}]`

$$\text{Out[43]} = -\frac{1}{n^3 \pi^3} 80 (n \pi + n \pi \text{Cos}[n \pi] - 2 \text{Sin}[n \pi])$$

$$\text{Out[44]} = \frac{20}{3}$$


In[45]:= `T[m1_, m2_, x_, t_] :=  
c0 + Sum[a[n] Exp[-\alpha n^2 Pi^2 t / L^2] Cos[n Pi x / L], {n, m1, m2}]`

```
In[46]:= Plot3D[N[T[1, 8, x, t]], {x, 0, 1}, {t, 0, 600},
  PlotRange -> All,
  AxesLabel -> {"Ort x in m ", "Zeit t in s ", "Temperatur u in C° "},
  AxesStyle -> Directive[Black, Bold, 12], ViewPoint -> {-2, 2, 3}]
```




Man erkennt einen weitgehenden Temperatenausgleich nach circa 10 Minuten im ganzen Stab. Die Temperatur geht mit wachsender Zeit überall gegen den Mittelwert  $c_0$  von  $f$ , wie man der Lösungsreihe ansieht. Hier die Temperatur nach 600 s (rot) und der Mittelwert von  $f$  (blau):

```
In[47]:= p1 := Plot[T[1, 8, x, 600], {x, 0, L}, PlotStyle -> Directive[
  Red, Thickness[0.005]], PlotRange -> All]
p2 := Plot[c0, {x, 0, L}, PlotStyle -> Directive[
  Blue, Thickness[0.005]], PlotRange -> All]
Show[p1, p2]
```



Nachfolgend eine mit *Mathematica* einfach mögliche Animation zum Temperatenausgleich im zeitlichen Verlauf über 2 Minuten. Da wir mit einer Näherungssumme für die Anfangsbedingung rechnen, ist die Randtemperatur zu Beginn nicht exakt Null, da eine Partialsomme für den 2-periodisch gerade fortgesetzten Parabelbogen nie ganz "in die Ecken hineinreicht".

```
In[50]:= Animate[Plot[N[T[1, 8, x, t]], {x, 0, 1}, PlotRange -> {0, 10},
  PlotStyle -> Directive[Red, Thickness[0.015]], ImageSize -> Small], {t, 0, 120},
  AnimationRepetitions -> 1, AnimationRate -> 1, RefreshRate -> 1]
```



Hier schließlich noch kurz ein Test, dass alles stimmt.  
 Jede Partialsumme der Fourierreihen-Lösung erfüllt nach  
 Konstruktion die Randbedingungen :

```
In[51]:= test = FullSimplify[ $\partial_t T[1, 8, x, t] - \alpha \partial_{x,x} T[1, 8, x, t]$ ]
```

Out[51]= 0

```
In[52]:= D[T[1, 8, x, t], x] /. x -> 0
```

Out[52]= 0

```
In[53]:= D[T[1, 8, x, t], x] /. x -> L
```

Out[53]= 0

Anmerkung: Beide Beispiele sind "unphysikalische" mathematische Modelle, in denen kein Wärmehaushalt mit der Umgebung berücksichtigt ist, der unter realen Verhältnissen eine Rolle spielen würde.

### Zum Abschluss noch der Fall eines inhomogenen Problems

Wir betrachten nun noch ein inhomogenes Problem. Als rechte Seite  $G$  der Gleichung  $\partial_t U[x, t] - \alpha \partial_{x,x} U[x, t] = G[x, t]$  wählen wir eine zeitlich konstante Wärmestromdichte, so dass  $G[x, t] = 0.6 (\text{UnitStep}[x-L/4] - \text{UnitStep}[x-3/4 L])$  (in  $C^0/s$ ) ist. Außerdem wählen wir als Anfangsbedingung  $U[x, 0] = 0$  für  $x$  in  $[0, L]$  und wie zuletzt Neumannsche Randbedingungen. Das Modell beschreibt ein um die Stabmitte herum gleichmäßiges Aufheizen unseres Kupferstabs, der ansonsten als perfekt isoliert gedacht wird. (Temperatur jetzt als  $U$  bezeichnet, damit es -solange weiter oben die Animation noch läuft- keinen Bezeichnungskonflikt gibt.)

Wir beginnen mit einem Ansatz mittels "Variation der Konstanten", d.h. wir verwenden den Lösungsansatz  $U[x, t] = c_0[t] + \sum_{n=1}^{\infty} c_n[t] \text{Cos}[n\pi x/L]$ , setzen diesen in die Gleichung ein und

verwenden die Rand- und Anfangsbedingungen (zur Übung bitte durchführen).  
 Damit ergibt sich  $c_n'[t] + (\alpha n^2 \pi^2) / L^2 c_n[t] = g_n$  für  $n > 0$ ,  $c_0[t] = g_0 t$ .  
 Dabei bezeichnen  $g_n$  die Fourier-Cosinuskoeffizienten der Inhomogenität  $G$ ,  $g_0$  den Mittelwert von  $G$ . Berechnung für  $G[x,t]$  mit  $L=1$  m und  $\alpha:=117 \cdot 10^{(-6)} \text{ m}^2/\text{s}$  wie oben und ergibt mit  $c_0[t] = g_0 t$  (aus  $c_0'[t] = g_0$  und  $c_0[0]=0$ )

```
In[54]:= L = 1; G[x_] := 6 / 10 (UnitStep[x - L / 4] - UnitStep[x - 3 / 4 L]);
g[n_] = 2 / L Integrate[G[x] Cos[n Pi x / L], {x, 0, L}]
g_0 = 1 / L Integrate[G[x], {x, 0, L}]
```

$$\text{Out[55]} = \frac{6 \left( -\text{Sin}\left[\frac{n \pi}{4}\right] + \text{Sin}\left[\frac{3 n \pi}{4}\right] \right)}{5 n \pi}$$

$$\text{Out[56]} = \frac{3}{10}$$

```
In[57]:= sol = DSolve[{D[c[n, t], {t, 1}] + \alpha n^2 Pi^2 / L^2 c[n, t] == g[n],
c[n, 0] == 0}, c[n, t], {n, t}]
```

$$\text{Out[57]} = \left\{ \left\{ c[n, t] \rightarrow -\frac{1}{39 n^3 \pi^3} 400\,000 e^{-\frac{117 n^2 \pi^2 t}{1\,000\,000}} \left( -1 + e^{\frac{117 n^2 \pi^2 t}{1\,000\,000}} \right) \left( \text{Sin}\left[\frac{n \pi}{4}\right] - \text{Sin}\left[\frac{3 n \pi}{4}\right] \right) \right\} \right\}$$

```
In[58]:= ckoeff[n_, t_] = sol[[1, 1, 2]]
```

$$\text{Out[58]} = -\frac{1}{39 n^3 \pi^3} 400\,000 e^{-\frac{117 n^2 \pi^2 t}{1\,000\,000}} \left( -1 + e^{\frac{117 n^2 \pi^2 t}{1\,000\,000}} \right) \left( \text{Sin}\left[\frac{n \pi}{4}\right] - \text{Sin}\left[\frac{3 n \pi}{4}\right] \right)$$

Um ein Gefühl zu erhalten, hier einige der Fourierkoeffizienten der Lösung zu den Zeiten 1, 31, 61 ... bis 151 s in 30 s Abständen. (Wählen Sie weitere Beispiele.) Überlegen Sie daran, wie sich in etwa die Gestalt der Lösung zeitlich entwickelt. Es ist durchaus lehrreich, solche Zahlenentwicklungen immer wieder in konkreten Beispielen "unter die Lupe zu nehmen", um mathematisch-physikalisches Gespür auszubilden!

```
In[70]:= Table[N[ckoeff[n, t]], {n, 2, 26, 4}, {t, 1, 151, 30}]
```


```
Out[70]= {{-0.381091, -11.0324, -20.3055, -28.3787, -35.4072, -41.5263},
{0.124714, 2.21862, 2.82026, 2.99313, 3.0428, 3.05707},
{-0.0721486, -0.643123, -0.660993, -0.661552, -0.661569, -0.66157},
{0.048833, 0.240881, 0.241097, 0.241097, 0.241097, 0.241097},
{-0.0354059, -0.113437, -0.113438, -0.113438, -0.113438, -0.113438},
{0.026602, 0.0621309, 0.0621309, 0.0621309, 0.0621309, 0.0621309},
{-0.0203964, -0.0376405, -0.0376405, -0.0376405, -0.0376405, -0.0376405}}
```

Wir betrachten eine Partialsumme der Fourierreihe der exakten Lösung für das Problem und plotten diese. **Nur jeder vierte Fourierkoeffizient ist ungleich Null.**

Um die stufenförmige Inhomogenität gut nachzubilden, wählen wir eine höhere Ordnung ( $m=30$ ) des trigonometrischen Polynoms zur Näherung der Wärmestromdichte und eine entsprechende Ordnung des trigonometrischen Näherungspolynoms für die Lösung. Die "Stufenform" der Inhomogenität bleibt anfänglich in der Lösung eine ganze Weile lang weitgehend erhalten bevor der Wärmeausgleich greift.


```
In[60]:= U[m_, x_, t_] := g_0 t + Sum[ckoeff[n, t] Cos[n Pi x / L], {n, 1, m}]
```

```
In[61]:= Plot3D[N[U[30, x, t]], {x, 0, L}, {t, 0, 600},
  PlotRange -> All, AxesLabel -> {"Ort x in m ", " Zeit t in s ", "U in C° "},
  AxesStyle -> Directive[Black, Bold, 12], ViewPoint -> {-2, 2, 3},
  Ticks -> {{0.0, 0.5, 0.75}, {10, 60, 300, 600}, {0, 100, 200, 300}}]
```



Hier wieder der zeitliche Temperaturverlauf als Animation (man beobachtet wegen der Näherung von  $G$  durch ein trigonometrisches Polynom negative Temperaturen  $U$  am Rand und eine "Welligkeit". Beide Effekte sind auf das **Gibbs-Phänomen** zurückzuführen.) Eine mathematisch präzise Behandlung von Differentialgleichungen bei unstetigen rechten Seiten wie hier  $G$  gelingt im Rahmen der Distributionentheorie (vgl. [1]).

```
In[62]:= Animate[Plot[N[U[30, x, t]], {x, 0, 1}, PlotRange -> {-2, 260},
  PlotStyle -> Directive[Red, Thickness[0.015]], ImageSize -> Small], {t, 0, 600},
  AnimationRepetitions -> 1, AnimationRate -> 1, RefreshRate -> 1]
```



Hier noch eine Darstellung der durch arithmetische Mittelung der Partialsummen des Ergebnisses geglätteten Temperaturentwicklung in einiger Entfernung vom Bereich  $[L/4, 3L/4]$ , der direkt erwärmt wird. Wir sehen einen in etwa linearen Anstieg der


Temperatur, der stärker wird, wenn man näher an das Intervall, welches erwärmt wird, heranrückt. (Testen Sie es selbst.)

In[63]=

```
Ugeglaettet[m_, x_, t_] :=
  g_0 t + Sum[ckoeff[n, t] (1 - n / (m + 1)) Cos[n Pi x / L], {n, 1, m}]
```

In[64]=

```
Plot[N[Ugeglaettet[30, L / 5, t]], {t, 0, 600}, PlotRange -> {0, 200},
  PlotStyle -> Directive[Red, Thickness[0.008]]]
```



Und der Schlusstest, dass alles stimmt: Die mittlere Temperatur nach 600 s entspricht der zugeführten Wärme entsprechend  $0.3 \text{ C}^0/\text{s}$  im Mittel für den ganzen Stab mit der angenommenen perfekten Isolierung und die Differentialgleichung wie auch die Anfangs- und die Randbedingungen sind erfüllt:

In[65]=  $1 / L \text{ Integrate}[U[30, x, 600], \{x, 0, L\}]$

Out[65]= 180

In[66]=

```
test = FullSimplify[
  D_t U[30, x, t] - alpha D_x,x U[30, x, t] - (g_0 + Sum[g[n] Cos[n Pi x / L], {n, 1, 30}]]]
```

Out[66]= 0

In[67]=

```
D[U[30, x, t], x] /. x -> 0
```

Out[67]= 0

In[68]=

```
D[U[30, x, t], x] /. x -> L
```

Out[68]= 0

In[69]=

```
U[30, x, 0]
```

Out[69]= 0

Literaturhinweise : Bei mehr Interesse an partiellen Differentialgleichungen empfehle ich

[2] M. M. Smirnov Second-Order Partial Differential Equations, Noordhoff, Groningen, 1966

[3] W. A. Strauss Partielle Differentialgleichungen, Vieweg, Wiesbaden, 1996